The Mathematica® Journal

A General Method for Constructing Ramanujan-Type Formulas for Powers of $1/\pi$

N. D. Bagis

This article discusses the theoretical background for generating Ramanujan-type formulas for $1/\pi^p$ and constructs series for $1/\pi^4$ and $1/\pi^6$. We also study the elliptic alpha function, whose values are useful for such evaluations.

Introduction

The standard definitions of the complete elliptic integrals of the first and second kind (see [1], [2], [3], [4]) are respectively:

$$K(x) = \int_0^{\pi/2} \frac{1}{\sqrt{1 - x^2 \sin(t)^2}}\, dt, \qquad E(x) = \int_0^{\pi/2} \sqrt{1 - x^2 \sin(t)^2}\, dt. \tag{1}$$

In *Mathematica*, these are `EllipticK[x^2]` and `EllipticE[x^2]`.

We also have

$$K(x) = \frac{\pi}{2}\, {}_2F_1\left(1/2, 1/2, 1; x^2\right) = \frac{\pi}{2} \sum_{n=0}^{\infty} \frac{(1/2)_n (1/2)_n}{(1)_n} \frac{x^{2n}}{n!} \tag{2}$$

and (see [5], [6]):

$$\dot{K}(x) = \frac{d\,K(k)}{d\,k} = \frac{E(k)}{k\left(1 - k^2\right)} - \frac{K(k)}{k}. \tag{3}$$

The elliptic singular moduli k_r is defined to be the solution of the equation

$$\frac{K\left(\sqrt{1-w^2}\right)}{K(w)} = \sqrt{r}\,. \tag{4}$$

In *Mathematica*, k_r is computed using `w = k[r]^2 = InverseEllipticNomeQ[E^(-Pi Sqrt[r])]`.

The complementary modulus is given by $(k'_r)^2 = 1 - k_r^2$. (For evaluations of k_r see [7], [8], [9]).

We need the following relation satisfied by the elliptic alpha function (see [7]):

$$a(r) = \frac{\pi}{4\,K\,(k_r)^2} - \sqrt{r}\left(\frac{E(k_r)}{K(k_r)} - 1\right). \tag{5}$$

Our method requires finding derivatives of powers of the elliptic integrals K and E that can always be expressed in terms of K, k_r, and $a(r)$. This article uses *Mathematica* to carry out these evaluations.

The function $a(r)$ is not widely known (see [7, 10]). Like the singular moduli, the elliptic alpha function can be evaluated from modular equations. The case $a(4\,r)$ is given in [7] Chapter 5:

$$a(4\,r) = (1+k_r)^2\,a(r) - 2\sqrt{r}\;k_r. \tag{6}$$

In view of [7], [11], and [5], the formula for $a(9\,r)$ is

$$\frac{a\,(9\,r)}{\sqrt{r}} - k_{9\,r}^2 = 1 - \frac{k_{9\,r}\,k_r}{3\,M_{3\,r}} - \frac{k'_{9\,r}\,k'_r}{3\,M_{3\,r}} - \frac{1}{3\,M_{3\,r}} - \frac{1}{3\,M_{3\,r}^2} + \frac{1}{M_{3\,r}^2}\left(\frac{a(r)}{\sqrt{r}} - \frac{k_r^2}{3}\right), \tag{7}$$

where $M_{3\,r}$ is a root of the polynomial equation

$$27\,M_{3\,r}^4 - 18\,M_{3\,r}^2 - 8\left(1 - 2\,k_r^2\right)M_{3\,r} - 1 = 0. \tag{8}$$

In the next section, we review and extend the method for constructing a series for π^{-p} based on $a(r)$. These Ramanujan-type formulas for $1/\pi^p$, $p \geq 4$ are presented here for the first time. The only formulas that were previously known are of orders 1, 2, and 3 ([12], [13]). There are few general formulas of order 2 and only one for order 3, due to B. Gourevitch (see references [14], [15], [5], [16], [17], [18]:

$$\sum_{n=0}^{\infty}\binom{2\,n}{n}^7 \frac{1}{2^{20\,n}}\left(1 + 14\,n + 76\,n^2 + 168\,n^3\right) = \frac{32}{\pi^3}\,. \tag{9}$$

In the last section we prove a formula for the evaluation of $a(25\,r)$ in terms of $a(r)$.

The General Method and the Construction of Formulas for $1/\pi^4$ and $1/\pi^6$

We have (see [16]):

$$\phi(x) = {}_3F_2\left(\frac{1}{2}, \frac{1}{2}, \frac{1}{2}; 1, 1; x\right) = \frac{4\,K^2\left(\sqrt{\frac{1}{2}\left(1-\sqrt{1-x}\right)}\right)}{\pi^2}. \tag{10}$$

This is the *Mathematica* definition.

```
ϕ[x_] := 4 EllipticK[1/2 (1 - Sqrt[1 - x])]^2 / π^2
```

Define $c_p(n)$, $p = 2, 4, 6, \ldots$, such that

$$\phi^p(x) = \left(\sum_{n=0}^{\infty} \frac{\binom{2n}{n}^3}{64^n} x^n\right)^p = \sum_{n=0}^{\infty} c_p(n)\, x^n. \tag{11}$$

It turns out that

$$\begin{aligned} c_2(n) &= \frac{1}{2^{6n}} \sum_{s=0}^{n} \binom{2s}{s}^3 \binom{2n-2s}{n-s}^3, \\ c_4(n) &= \sum_{s=0}^{n} c_2(s)\, c_2(n-s), \\ c_6(n) &= \sum_{s=0}^{n} c_4(s)\, c_2(n-s), \\ &\ldots. \end{aligned} \tag{12}$$

Here are the *Mathematica* definitions for $c_p(n)$ for $p = 2, 4, 6$.

```
c2[n_] :=
  1/2^(6 n) Sum[Binomial[2 s, s]^3 Binomial[2 n - 2 s, n - s]^3, {s, 0, n}]

c4[n_] := Sum[c2[n - s] c2[s], {s, 0, n}]
```

```
c6[n_] := Sum[c4[n - s3] c2[s3], {s3, 0, n}]
```

Consider the following equation for the function $\phi^p(x)$:

$$A_p\, x^p \frac{d^p\, \phi(x)^p}{dx^p} + A_{p-1}\, x^{p-1} \frac{d^{p-1}\, \phi(x)^p}{dx^{p-1}}(x) + \ldots + A_1\, x \frac{d\, \phi(x)^p}{dx} + A_0\, \phi(x)^p\,(x) = \sum_{n=0}^{\infty} c_p(n)\, x^n \left(C_p\, n^p + C_{p-1}\, n^{p-1} + \ldots + C_1\, n + C_0\right) = \frac{g}{\pi^p}. \tag{13}$$

Set $x = 1 - (1 - 2\,w)^2$; then $w = k_r^2$ and g, for suitable values of A_j, is a function of k_r and $a(r)$, so g is an algebraic number when $r \in N$. The A_j and C_j can be evaluated from (13). Higher values of $r \in N$ and k_r give more accurate and faster formulas for $1/\pi^4$ and $1/\pi^6$.

□ Series for $1/\pi^4$

The general formula produced by our method for $1/\pi^4$ is

$$\sum_{n=0}^{\infty} c_4(n)\, (k_r\, k'_r)^{2\,n}\Big[A_4\, n^4 + (A_3 - 6\,A_4)\, n^3 + (A_2 - 3\,A_3 + 11\,A_4)\, n^2 + (A_1 - A_2 + 2\,A_3 - 6\,A_4)\, n + A_0\Big] = \frac{g}{\pi^4}. \tag{14}$$

This computes the polynomial in the variable n in the sum (13).

```
npoly[p_, A_] :=
 A[0] + Collect[Table[Product[n - k, {k, 0, K}], {K, 0, -1 + p}].
    Array[A, p], n]

npoly[4, A] // TraditionalForm
```

$$A(4)\, n^4 + (A(3) - 6\, A(4))\, n^3 + (A(2) - 3\, A(3) + 11\, A(4))\, n^2 + (A(1) - A(2) + 2\, A(3) - 6\, A(4))\, n + A(0)$$

To find the A_j, the function `Arules` replaces `EllipticK[w]` by `EK[w]` and `EllipticE[w]` by `EE[w]` and sets all the Taylor expansion coefficients with respect to `EK[w]` to 0.

```
EE[w_] := (π - 4 a EK[w]^2 + 4 Sqrt[r] EK[w]^2)/(4 Sqrt[r] EK[w])
```

```
Arules[M_, p_] := (* Replace EllipticK[w] and
  EllipticE[w] by EE[w] and set all the Taylor
  expansion coefficients with respect to EK[w] to be 0. *)
 First@Simplify@Solve[
    Prepend[
     Table[
      0 == SeriesCoefficient[
        PowerExpand[
           Table[x^i D[ϕ[x]^p, {x, i}], {i, 0, p}].
             Array[A, p + 1, 0] /. x → (1 - (1 - 2 w)^2)
          ] /. EllipticK[w] → EK[w] /. EllipticE[w] → EE[w],
        {EK[w], 0, j}
       ],
      {j, 1, M}
     ],
     A[0] == 1],
    Array[A, p + 1, 0]
   ]
```

Choose `M` large enough to get a solution for all the `A[i]` for $p = 4$. (Here $a = a(r)$ and $w = k_r^2$.)

```
(Arules4 = Arules[20, 4]) // TraditionalForm
```

$$\Big\{A(0) \to 1,\ A(1) \to \Big(-210\sqrt{r}\,(2w-1)\,a^3 + 90\,r\,(20w^2 - 13w + 1)\,a^2 - 10\,r^{3/2}\,(304w^3 - 276w^2 + 57w - 2)\,a + r^2\,(2080w^4 - 2640w^3 + 995w^2 - 115w + 2)\Big)\Big/ \Big(2\,\big(105a^4 - 420\sqrt{r}\,w\,a^3 + 90\,r\,w\,(8w-1)\,a^2 - 20\,r^{3/2}\,w\,(32w^2 - 12w + 1)\,a + r^2\,w\,(256w^3 - 192w^2 + 43w - 2)\big)\Big),$$

$$A(2) \to \Big(r\,\big(45\,a^2\,(1-2w)^2 - 30\,a\sqrt{r}\,(24w^3 - 30w^2 + 11w - 1) + r\,(880w^4 - 1400w^3 + 735w^2 - 140w + 7)\big)\Big)\Big/ \Big(105a^4 - 420\sqrt{r}\,w\,a^3 + 90\,r\,w\,(8w-1)\,a^2 - 20\,r^{3/2}\,w\,(32w^2 - 12w + 1)\,a + r^2\,w\,(256w^3 - 192w^2 + 43w - 2)\Big),$$

$$A(3) \to \Big(2\,r^{3/2}\,(1-2w)^2\,\big(a\,(5 - 10w) + \sqrt{r}\,(28w^2 - 23w + 3)\big)\Big)\Big/ \Big(105a^4 - 420\sqrt{r}\,w\,a^3 + 90\,r\,w\,(8w-1)\,a^2 - 20\,r^{3/2}\,w\,(32w^2 - 12w + 1)\,a + r^2\,w\,(256w^3 - 192w^2 + 43w - 2)\Big),$$

$$A(4) \to \Big(r^2\,(1-2w)^4\Big)\Big/\Big(105a^4 - 420\sqrt{r}\,w\,a^3 + 90\,r\,w\,(8w-1)\,a^2 - 20\,r^{3/2}\,w\,(32w^2 - 12w + 1)\,a + r^2\,w\,(256w^3 - 192w^2 + 43w - 2)\Big)\Big\}$$

Now that we have the `A[i]`, this computes the sum on the left-hand side of (13).

```
lhs4[T_] := Sum[c4[n] x^n npoly[4, A], {n, 0, T}] /. Arules4 /.
  x → 1 - (1 - 2 w)^2
```

This computes the right-hand side of (13).

```
rhs4[a_, r_, w_] :=
 Module[{x = Denominator[A[1] /. Arules4]},
  (x /. r → 0 /. a → 1) / (π^4 x)]
```

```
rhs4[a, r, w] // TraditionalForm
```

$$105/\left(\pi^4\left(105\,a^4-420\sqrt{r}\;w\,a^3+90\,r\,w\,(8\,w-1)\,a^2-20\,r^{3/2}\,w\left(32\,w^2-12\,w+1\right)a+r^2\,w\left(256\,w^3-192\,w^2+43\,w-2\right)\right)\right)$$

Example 1. From [19] and [7], for $p=4$ and $r=2$, we have $k_2=\sqrt{2}-1$ and $a(2)=\sqrt{2}-1$. Hence we get the formula

$$\sum_{n=0}^{\infty}c_4(n)\left(-56+40\sqrt{2}\right)^n\left[1+5\left(\frac{292\,072+56\,267\sqrt{2}}{462\,719}\right)n+6\left(\frac{268\,641+81\,580\sqrt{2}}{462\,719}\right)n^2+4\left(\frac{134\,444+32\,155\sqrt{2}}{462\,719}\right)n^3-4\left(\frac{36\,209+34\,800\sqrt{2}}{462\,719}\right)n^4\right]=\frac{105}{\left(229\,441-162\,240\sqrt{2}\right)\pi^4}. \quad (15)$$

We verify this numerically.

```
verify1 = {r → 2, a → √2 - 1, w → (√2 - 1)^2};
```

```
rhs4[a, r, w] /. verify1 // FullSimplify
```

$$-\frac{105}{\left(-229\,441+162\,240\sqrt{2}\right)\pi^4}$$

```
Array[A, 5, 0] /. Arules4 /. verify1 // FullSimplify
```

$$\left\{1, \frac{3465146 + 760235\sqrt{2}}{462719}, \frac{2211322 - 99060\sqrt{2}}{462719}, -\frac{980\left(338 + 721\sqrt{2}\right)}{462719}, -\frac{4\left(36209 + 34800\sqrt{2}\right)}{462719}\right\}$$

```
N[lhs4[100] - rhs4[a, r, w] /. verify1]
```

9.10383×10^{-15}

Example 2. Here is another example for $p = 4$ that we verify numerically.

$$\text{verify2} = \left\{r \to 15,\ a \to -\frac{1}{2} + \sqrt{5 - \frac{5\sqrt{3}}{2}},\ w \to \frac{1}{8\left(376 + 168\sqrt{5} + \sqrt{6\left(47067 + 21049\sqrt{5}\right)}\right)}\right\};$$

```
rhs4[a, r, w] /. verify2 // FullSimplify
```

$$\frac{1792}{17\,\pi^4}$$

```
Array[A, 5, 0] /. Arules4 /. verify2 // FullSimplify
```

$$\left\{1, \frac{3}{544}\left(8595 + 36049\sqrt{5}\right), \frac{135}{544}\left(3633 + 2347\sqrt{5}\right), \frac{45}{272}\left(4635 + 2089\sqrt{5}\right), \frac{135}{544}\left(487 + 189\sqrt{5}\right)\right\}$$

```
N[lhs4[100] - rhs4[a, r, w] /. verify2]
```

-2.19824×10^{-14}

□ Series for $1/\pi^6$

The coefficients of A_j and the parameters for the $1/\pi^6$ formula are obtained using the same method as for $1/\pi^4$. (The same can be done as well for $1/\pi^2$, of course.) Higher values of $r \in N$ and k_r give more accurate and faster formulas for $1/\pi^4$ and $1/\pi^6$.

For $p = 6$ we get

$$\sum_{n=0}^{\infty} c_6(n)\,(k_r\,k'_r)^{2n}\big[A_6\,n^6 + (A_5 - 15\,A_6)\,n^5 + (A_4 - 10\,A_5 + 85\,A_6)\,n^4 + (A_3 - 6\,A_4 + 35\,A_5 - 225\,A_6)\,n^3 + (A_2 - 3\,A_3 + 11\,A_4 - 50\,A_5 + 274\,A_6)\,n^2 + (A_1 - A_2 + 2\,A_3 - 6\,A_4 + 24\,A_5 - 120\,A_6)\,n + A_0\big] = \frac{g}{\pi^6}. \quad (16)$$

This calculates the A_j.

```
Arules6 = Arules[20, 6];
```

Example 3. For $r = 2$,

$$\sum_{n=0}^{\infty} c_6(n)\left(-56 + 40\sqrt{2}\right)^n\Big[1 + \frac{28\,335\,508\,172 - 240\,070\,543\sqrt{2}}{12\,623\,771\,801}\,n + \frac{22\,911\,684\,702 - 3\,047\,538\,900\sqrt{2}}{12\,623\,771\,801}\,n^2 + \frac{6\,110\,502\,200 - 5\,456\,734\,120\sqrt{2}}{12\,623\,771\,801}\,n^3 - \frac{1\,196\,112\,280 + 3\,649\,618\,320\sqrt{2}}{12\,623\,771\,801}\,n^4 - \frac{505\,494\,672 + 788\,011\,092\sqrt{2}}{12\,623\,771\,801}\,n^5 + \frac{463\,408\,744 + 244\,639\,040\sqrt{2}}{12\,623\,771\,801}\,n^6\Big] = \frac{3465}{\left(629\,823\,301 - 445\,352\,320\sqrt{2}\right)\pi^6}. \quad (17)$$

Example 4. For $r = 7$, we have $k_r^2 = \frac{8-3\sqrt{7}}{16}$ and $a(7) = \frac{\sqrt{7}-2}{2}$; then

$$\sum_{n=0}^{\infty} \frac{c_6(n)}{64^n} \left[1 + \frac{913\,150}{307\,323} n - \frac{75\,313}{102\,441} n^2 - \frac{4\,998\,980}{307\,323} n^3 - \frac{1\,126\,755}{34\,147} n^4 - \frac{1\,080\,450}{34\,147} n^5 - \frac{453\,789}{34\,147} n^6\right] = -\frac{14\,417\,920}{34\,147\,\pi^6}. \tag{18}$$

We verify this numerically.

```
verify3 = {r → 7, a → (√7 - 2)/2, w → (8 - 3 √7)/16};
```

```
Array[A, 7, 0] /. Arules6 /. verify3 // FullSimplify
```

$$\left\{1, -\frac{28\,260\,715}{307\,323}, -\frac{39\,852\,925}{34\,147}, -\frac{676\,514\,090}{307\,323}, -\frac{41\,427\,540}{34\,147}, -\frac{7\,887\,285}{34\,147}, -\frac{453\,789}{34\,147}\right\}$$

```
rhs6[a_, r_, w_] :=
 Module[{x = Denominator[A[1] /. Arules6]},
  (x /. r → 0 /. a → 1) / (π^6 x)]
```

```
rhs6[a, r, w] // TraditionalForm
```

$$3465\Big/\Big(\pi^6 \big(3465\,a^6 - 20\,790\sqrt{r}\,w\,a^5 + 4725\,r\,w\,(12\,w-1)\,a^4 - 1260\,r^{3/2}\,w\left(72\,w^2 - 18\,w + 1\right)a^3 + 105\,r^2\,w\left(864\,w^3 - 432\,w^2 + 65\,w - 2\right)a^2 - 21\,r^{5/2}\,w\left(2592\,w^4 - 2160\,w^3 + 616\,w^2 - 59\,w + 1\right)a + r^3\,w\left(15\,552\,w^5 - 19\,440\,w^4 + 8944\,w^3 - 1700\,w^2 + 110\,w - 1\right)\big)\Big)$$

```
rhs6[a, r, w] /. verify3 // FullSimplify
```

$$-\frac{14\,417\,920}{34\,147\,\pi^6}$$

```
lhs6[T_] := Sum[c6[n] x^n npoly[6, A], {n, 0, T}] /. Arules6 /.
  x → 1 - (1 - 2 w)^2
```

```
N[lhs6[50] - rhs6[a, r, w] /. verify3]
```

```
1.44329 × 10^-15
```

Example 5. For $r = 15$, we have $k_r^2 = \frac{\left(2-\sqrt{3}\right)^2 \left(\sqrt{5}-\sqrt{3}\right)^2 \left(3-\sqrt{5}\right)^2}{128}$ and $a(15) = \frac{\sqrt{15}-\sqrt{5}-1}{2}$; then

$$\sum_{n=0}^{\infty} c_6(n) \left(\frac{47-21\sqrt{5}}{128}\right)^n \left[1 + \left(\frac{2\,877\,117\,109\,830 + 924\,178\,552\,332\sqrt{5}\; n}{293\,049\,243\,769}\right) + \left(\frac{15\,689\,590\,644\,975 + 6\,660\,423\,786\,240\sqrt{5}}{293\,049\,243\,769}\right) n^2 + \left(\frac{51\,863\,088\,153\,600 + 23\,066\,524\,139\,820\sqrt{5}}{293\,049\,243\,769}\right) n^3 + \left(\frac{106\,483\,989\,569\,175 + 47\,630\,637\,457\,200\sqrt{5}}{293\,049\,243\,769}\right) n^4 + \left(\frac{130\,261\,549\,416\,750 + 58\,266\,415\,341\,540\sqrt{5}}{293\,049\,243\,769}\right) n^5 + \left(\frac{75\,619\,648\,012\,725 + 33\,817\,435\,224\,300\sqrt{5}}{293\,049\,243\,769}\right) n^6\right] = \frac{20\,185\,088}{\left(11\,556\,387 - 5\,162\,500\sqrt{5}\right)\pi^6}. \tag{19}$$

Evaluating the Elliptic Alpha Function $a(25\,r)$

It is clear from the results in the previous section that getting rapidly convergent series for $1/\pi$ and its even powers requires values of the alpha function $a(r)$ for large $r \in N$, say $r = 5000$ (see [14], [20], [5]). In this section we address this problem.

From (4), (7), and [2] pages 121–122, Chapter 21, if we set $y = \pi\sqrt{r}$, $q = e^{-\pi\sqrt{r}}$, $K(k_r) = K[r]$, $k'_r = \sqrt{1-k_r^2}$, then

$$1 - 24\sum_{n=1}^{\infty}\frac{n\,q^{2n}}{1-q^{2n}} = \frac{3}{\pi\sqrt{r}} + \left(1 + k_r^2 - \frac{3\,a(r)}{\sqrt{r}}\right)\frac{4}{\pi^2}K[r]^2. \tag{20}$$

From the duplication formula

$$k_{r/4} = \frac{2\sqrt{k_r}}{1+k_r}$$

and

$$a(4\,r) = (1+k_{4\,r})^2\,a(r) - 2\sqrt{r}\;k_{4\,r},$$

equation (20) becomes

$$1 - 24\sum_{n=1}^{\infty}\frac{n\,q^{n}}{1-q^{n}} = \frac{6}{\pi\sqrt{r}} + \left(1 + k_r^2 - \frac{6\,a(r)}{\sqrt{r}}\right)\frac{4}{\pi^2}K[r]^2. \tag{21}$$

Setting

$$T_{p,r} = \left(1 - 24\sum_{n=1}^{\infty}\frac{n\,q^{2n}}{1-q^{2n}}\right) - p\left(1 - 24\sum_{n=1}^{\infty}\frac{n\,q^{2pn}}{1-q^{2pn}}\right) \tag{22}$$

gives the following proposition.

Proposition 1

$$\frac{1}{m_p^2\sqrt{r}}\,a\left(p^2\,r\right) = -\frac{1+k_r^2}{3} + \frac{p\left(1+k_{p^2\,r}^2\right)}{3\,m_p^2} + \frac{\pi^2\,T_{p,r}}{12\,K[r]^2} + \frac{a(r)}{\sqrt{r}}. \tag{23}$$

This connects Ramanujan's results of Chapter 21 in [2] with the evaluation of the alpha function and the evaluations of π. Solving (23) with respect to $T_{p,r}$ gives

$$T_{p,r} = \frac{4\,K[r]^2}{\pi^2\sqrt{r}\;m_p^2}\left[3\,a\left(p^2\,r\right) - p\sqrt{r}\,\left(1 + k_{p^2\,r}^2\right) + \left(-3\,a(r) + \sqrt{r}\,\left(1 + k_r^2\right)\right) m_p^2\right].$$

Equations (21), (22), and (23) give another interesting formula,

$$1 - 24\sum_{n=1}^{\infty}\frac{n\,q^{2\,p\,n}}{1 - q^{2\,p\,n}} = \frac{3}{\pi\sqrt{r}\;p} + \frac{4\,K[r]^2}{\pi^2\sqrt{r}\;p\,m_p^2}\left[-3\,a\left(p^2\,r\right) + p\sqrt{r}\,\left(1 + k_{p^2\,r}^2\right)\right]$$

where

$$m_p = \frac{K[r]}{K\left[p^2\,r\right]}. \tag{24}$$

Entry 4 of [2], p. 436 is

$$T_{5,r} = \frac{\left(x^2 + 22\,q^2\,x\,y + 125\,q^4\,y^2\right)^{1/2}}{(x\,y)^{1/6}}, \tag{25}$$

where $x = f\left(-q^2\right)^6$ and $y = f\left(-q^{10}\right)^6$.

Set

$$A_r = \frac{1}{R\left(q^2\right)^5} - 11 - R\left(q^2\right)^5 = \frac{x}{q^2\,y}, \tag{26}$$

where $R(q)$ is the Rogers–Ramanujan continued fraction (see [2], [21], [22]):

$$R(q) = \frac{q^{1/5}}{1+}\,\frac{q}{1+}\,\frac{q^2}{1+}\,\frac{q^3}{1+}\,\ldots \tag{27}$$

and

$$f\left(-q^2\right)^6 = \frac{2\,k_r\,k'_r\,K[r]^3}{\pi^3\,q^{1/2}}; \tag{28}$$

this gives

$$T_{5,r} = 4\,\frac{(k_r\,k'_r)^{2/3}\sqrt{125 + 22\,A_r + A_r^2}}{6\sqrt[3]{2}\;A_r^{5/6}} = 2^{2/3}\,\frac{(k_r\,k'_r)^{2/3}\left[R\left(q^2\right)^{-5} + R\left(q^2\right)^5\right]}{3\left[R\left(q^2\right)^{-5} - 11 - R\left(q^2\right)^5\right]^{5/6}} \tag{29}$$

and hence the evaluation

$$-4-24\sum_{n=1}^{\infty}\frac{n\,q^{2n}}{1-q^{2n}}+120\sum_{n=1}^{\infty}\frac{n\,q^{10n}}{1-q^{10n}}=2^{2/3}\,\frac{(k_r\,k'_r)^{2/3}\left[R(q^2)^{-5}+R\left(q^2\right)^5\right]}{3\left[R\left(q^2\right)^{-5}-11-R\left(q^2\right)^5\right]^{5/6}}. \tag{30}$$

But for the evaluation of the Rogers–Ramanujan continued fraction, from [22] we have

Proposition 2

If $q=e^{-\pi\sqrt{r}}$ and r is a positive real, then

$$A_r=a_{4r}=\left(\frac{k_r\,k'_r}{w_r\,w'_r}\right)^2\left(\frac{w_r}{k_r}+\frac{w'_r}{k'_r}-\frac{w_r\,w'_r}{k_r\,k'_r}\right)^3 \tag{31}$$

with

$$\begin{aligned} m_5&=\frac{w_r}{k_r}+\frac{w'_r}{k'_r}-\frac{w_r\,w'_r}{k_r\,k'_r}\,,\\ w_r&=\sqrt{k_r\,k_{25r}}\,,\\ w'_r&=\sqrt{k'_r\,k'_{25r}}\,. \end{aligned} \tag{32}$$

Proposition 3

$$\frac{3\,a(25\,r)}{m_5^2}-\frac{3\,a(r)}{\sqrt{r}}= \frac{5}{m_5^2}\left(1+k_{25r}^2\right)-\left(1+k_r^2\right)-2^{2/3}\,A_r^{-5/6}(k_r\,k'_r)^{2/3}\sqrt{125+A_r+A_r^2}\,. \tag{33}$$

Proof

From (23), (28), and (31),

$$\frac{3\,a(25\,r)}{m_5^2\sqrt{r}}-\frac{3\,a(r)}{\sqrt{r}}= -1+\frac{5}{m_5^2}+\frac{5\,k_{25r}^2}{m_5^2}-k_r^2-\frac{2^{2/3}\sqrt{125+22\,A_r+A_r^2}\;(k_r\,k'_r)^{2/3}}{A_r^{5/6}}, \tag{34}$$

with (see [22])

$$A_{r/4}=\left(\frac{k'_r}{k'_{25r}}\right)^2\sqrt{\frac{k_r}{k_{25r}}}\;m_5^{-3}.\ \blacksquare \tag{35}$$

In some cases, the next formula from [9] is very useful:

$$k_{25^n r_0} = \sqrt{1/2 - 1/2\sqrt{1 - 4\,(k_{r_0}\,k'_{r_0})^2 \prod_{j=1}^{n} P^{(j)}\left(\sqrt[12]{\frac{k_{r_0}\,k'_{r_0}}{k_{r_0}\,k'_{r_0/25}}}\right)^{24}}}\,. \tag{36}$$

Here the function P is $P(x) = U\left(Q^{1/6}\left(U^{*6}(x)\right)\right)$, where U, U^*, and Q are as defined in [9] and $P^{(n)}(x) = P(P(\ldots P(x)\ldots))$ is the n^{th} iterate of P.

The coefficient $m_5 = 1/M_5$ was defined in (24) and occurred in (32); M_5 also satisfies the equation

$$(5\,M_5 - 1)^5\,(1 - M_5) = 256\,(k_r\,k'_r)\,M_5.$$

If we know k_r and $k_{25\,r}$, we can evaluate A_r from (31) and then we can evaluate $a(25\,r)$.

The following conjecture is most compactly expressed in terms of the quantity

$$Y_{\sqrt{-r}} = \frac{1}{6}\left(R\left(e^{-\pi\sqrt{r}}\right)^{-5} - 11 - R\left(e^{-\pi\sqrt{r}}\right)^{5}\right) = \frac{A_r}{6}. \tag{37}$$

The function j_r is the j-invariant (see [23], [8]). For more properties of j_r and A_r see [24].

Conjecture (Algorithm for A_r)

Numerical results calculated with *Mathematica* indicate that whenever $\gcd(r, 5) = 1$, then $\deg\left(Y_{\sqrt{-r/5}}\right) = \deg\left(j_{\sqrt{-r/5}}\right)$.

For a given $r \in N$ and $\deg\left(Y_{\sqrt{-r/5}}\right) = 2$, 4, or 8, if the smallest nested root of $j_{\sqrt{-r/5}}$ is $\sqrt{d}$, then we can evaluate the Rogers–Ramanujan continued fraction with integer parameters.

1. When $\deg\left(Y_{\sqrt{-r/5}}\right) = 2$,

$$Y_{\sqrt{-r/5}} = \frac{l + m\sqrt{d}}{t} \tag{38}$$

with $l^2 - m^2\,d = 1$, where l, m, d are positive integers.

2. When $\deg\left(Y_{\sqrt{-r/5}}\right) = 4$,

a) If $U \neq \frac{125}{64}$, then

$$Y_{\sqrt{-r/5}} = \frac{5}{8}\sqrt{a_0 + \sqrt{-1 + a_0^2}}\,\left(\sqrt{5 + p} - \sqrt{p}\right), \tag{39}$$

where

$$Y_{\sqrt{-r/5}}\, Y^{*}_{\sqrt{-r/5}} = \frac{125}{64}\left(a_0 + \sqrt{-1 + a_0^2}\right) \tag{40}$$

and where a_0 is the positive integer solution of $l^2 - m^2 d = 1$. Hence $l = a_0$ and $m = d^{-1/2}\sqrt{a_0^2 - 1}$ is a positive integer. The parameter p is a positive rational and can be found directly from the numerical value of $Y_{\sqrt{-r/5}}$.

b) If $U = \frac{125}{64}$, then

$$Y_{\sqrt{-r/5}} = A + \frac{1}{8}\sqrt{-125 + 64\, A^2}, \tag{41}$$

where we set $A = k + l\sqrt{d}$. Then a starting point for the evaluation of the integers k, l is

$$l^2 = \frac{(A-k)^2}{d}, \tag{42}$$

the square of an integer.

3. When $\deg\left(Y_{\sqrt{-1/20\, r}}\right) = 4$, then we can evaluate $Y_{\sqrt{-1/5\, r}}$.

The degree of $Y_{\sqrt{-1/5\, r}}$ is 8 and the minimal polynomial of $Y_{\sqrt{-1/5\, r}} \Big/ Y_{\sqrt{-1/20\, r}}$ is of degree 4 or 8 and symmetric. Hence the minimal polynomial can be reduced to at most a fourth-degree polynomial and so it is solvable. With the help of step 2, we can evaluate $Y_{\sqrt{-1/20\, r}}$.

$$Y_{\sqrt{-1/5\, r}} = \frac{5}{8}\sqrt{a_0 + \sqrt{-1 + a_0^2}}\,\left(\sqrt{p+5} - \sqrt{p}\right) 2^{-1}\left(\sqrt{x+4} - \sqrt{x}\right), \tag{43}$$

where $x = a_1 + b_1\sqrt{d} + c\sqrt{a_2 + b_2\sqrt{d}}$, a_1, b_1, a_2, b_2, c are integers, and

$$Y_{\sqrt{-1/20\, r}} = \frac{5}{8}\sqrt{a_0 + \sqrt{-1 + a_0^2}}\,\left(\sqrt{p+5} - \sqrt{p}\right). \tag{44}$$

Here are some values of $Y_{\sqrt{-1/5\, r}} = 1/8\, A_{r/5}$ that can found with the *Mathematica* built-in function `Recognize` or by solving Pell's equation and applying the conjecture.

$$Y_{\sqrt{-1/5}} = \frac{5\sqrt{5}}{8}. \tag{45}$$

$$Y_{\sqrt{-2/5}} = \frac{5}{8}\left(5 + 2\sqrt{5}\right). \tag{46}$$

$$Y_{\sqrt{-3/5}} = \frac{5}{16}\left(25 + 11\sqrt{5}\right). \tag{47}$$

$$Y_{\sqrt{-4/5}} = \frac{5}{16}\left(25 + 13\sqrt{5} + 5\sqrt{58 + 26\sqrt{5}}\right). \tag{48}$$

$$Y_{\sqrt{-5/5}} = \frac{125}{8}\left(2 + \sqrt{5}\right). \tag{49}$$

$$Y_{\sqrt{-6/5}} = \frac{5}{8}\left(50 + 35\sqrt{2} + 3\sqrt{5\left(99 + 70\sqrt{2}\right)}\right). \tag{50}$$

$$Y_{\sqrt{-9/5}} = \frac{5}{8}\left(225 + 104\sqrt{5} + 10\sqrt{1047 + 468\sqrt{5}}\right). \tag{51}$$

$$Y_{\sqrt{-12/5}} = \frac{5}{12}\left(1690 + 975\sqrt{3} + 29\sqrt{6755 + 3900\sqrt{3}}\right). \tag{52}$$

$$Y_{\sqrt{-14/5}} = \frac{5}{8}\left(1850 + 585\sqrt{10} + 7\sqrt{5\left(27\,379 + 8658\sqrt{10}\right)}\right). \tag{53}$$

$$Y_{\sqrt{-17/5}} = \frac{5}{8}\left(5360 + 585\sqrt{85} + 4\sqrt{3\,613\,670 + 391\,950\sqrt{85}}\right). \tag{54}$$

Example 6. If $r = 68 = 4 \times 17$, from (54) we have $d = 85$, hence

$$x = a_1 + b_1\sqrt{85} + c\sqrt{a_2 + b_2\sqrt{85}}, \tag{55}$$

$$Y_{\sqrt{-68/5}} \Big/ Y_{\sqrt{-17/5}} = 2^{-1}\left(\sqrt{x+4} - \sqrt{x}\right), \tag{56}$$

$a_1 = 2\,891\,581\,250$, $b_1 = 31\,363\,605$, $c = 12\,960$,
$a_2 = 99\,557\,521\,554$, $b_2 = 10\,798\,529\,365$.

Hence

$$Y_{\sqrt{-68/5}} = Y_{\sqrt{-17/5}}\, 2^{-1}\left(\sqrt{x+4} - \sqrt{x}\right) = \frac{5}{16} \left(5360 + 585\sqrt{85} + 4\sqrt{3\,613\,670 + 391\,950\sqrt{85}}\right)\left(\sqrt{x+4} - \sqrt{x}\right). \tag{57}$$

■ References

[1] M. Abramowitz and I. A. Stegun, *Handbook of Mathematical Functions*, New York: Dover, 1972.

[2] B. C. Berndt, *Ramanujan's Notebooks, Part III*, New York: Springer-Verlag, 1991.

[3] J. V. Armitage and W. F. Eberlein, *Elliptic Functions*, New York: New York: Cambridge University Press, 2006.

[4] E. T. Whittaker and G. N. Watson, *A Course of Modern Analysis*, 4th ed., Cambridge: Cambridge University Press, 1927.

[5] N. D. Bagis and M. L. Glasser, "Conjectures on the Evaluation of Alternative Modular Bases and Formulas Approximating $1/\pi$," *Journal of Number Theory*, **132**(10), 2012 pp. 2353–2370.

[6] N. D. Baruah and B. C. Berndt, "Eisenstein Series and Ramanujan-Type Series for $1/\pi$," *Ramanujan Journal*, **23**(1–3), 2010 pp. 17–44. link.springer.com/article/10.1007/s11139-008-9155-8.

[7] J. M. Borwein and P. B. Borwein, *Pi and the AGM*, New York: Wiley, 1987.

[8] D. Broadhurst, "Solutions by Radicals at Singular Values k_N from New Class Invariants for $N = 3 \bmod 8$." arxiv.org/abs/0807.2976.

[9] N. Bagis, "Evaluation of Fifth Degree Elliptic Singular Moduli." arxiv.org/abs/1202.6246v1.

[10] E. W. Weisstein, "Elliptic Alpha Function" from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/EllipticAlphaFunction.html.

[11] J. M. Borwein and P. B. Borwein, "A Cubic Counterpart of Jacobi's Identity and the AGM," *Transactions of the American Mathematical Society*, **323**(2), 1991 pp. 691–701. www.ams.org/journals/tran/1991-323-02/S0002-9947-1991-1010408-0/S0002-9947-1991-1010408-0.pdf.

[12] N. D. Baruah and B. C. Berndt, "Ramanujan's Series for $1/\pi$ Arising from His Cubic and Quartic Theories of Elliptic Functions," *Journal of Mathematical Analysis and Applications*, **341**(1), 2008 pp. 357–371. doi:10.1016/j.jmaa.2007.10.011.

[13] N. D. Baruah, B. C. Berndt, and H. H. Chan, "Ramanujan's Series for $1/\pi$: A Survey," *American Mathematical Monthly*, August–September, 2009 pp. 567–587. www.math.uiuc.edu/~berndt/articles/monthly567-587.pdf.

[14] N. Bagis, "Ramanujan-Type $1/\pi$ Approximation Formulas." arxiv.org/abs/1111.3139v1.

[15] S. Ramanujan, "Modular Equations and Approximations to π," *Quarterly Journal of Pure and Applied Mathematics*, **45**, 1914 pp. 350–372.

[16] W. Zudilin, "Ramanujan-Type Formulae for $1/\pi$: A Second Wind?" arxiv.org/abs/0712.1332.

[17] E. W. Weisstein, "Pi Formulas," from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/PiFormulas.html.

[18] The Mathematics Genealogy Project. "Jesús Guillera." (Jul 17, 2013) genealogy.math.ndsu.nodak.edu/id.php?id=124102.

[19] E. W. Weisstein, "Elliptic Lambda Function" from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/EllipticLambdaFunction.html.

[20] B. C. Berndt and H. H. Chan, "Eisenstein Series and Approximations to $1/\pi$," *Illinois Journal of Mathematics*, **45**, 2001 pp. 75–90. www.math.uiuc.edu/~berndt/publications.html.

[21] B. C. Berndt, *Ramanujan's Notebooks*, *Part V*, New York: Springer-Verlag, 1998.

[22] N. D. Bagis, "Parametric Evaluations of the Rogers–Ramanujan Continued Fraction," *International Journal of Mathematics and Mathematical Sciences*, #940839, 2011. doi:10.1155/2011/940839.

[23] B. C. Berndt and H. H. Chan, "Ramanujan and the Modular j-Invariant," *Canadian Mathematical Bulletin*, **42**(4), 1999 pp. 427–440. cms.math.ca/10.4153/CMB-1999-050-1.

[24] N. Bagis, "On a General Polynomial Equation Solved by Elliptic Functions." arxiv.org/abs/1111.6023v1.

N. D. Bagis, "A General Method for Constructing Ramanujan-Type Formulas for Powers of $1/\pi$," *The Mathematica Journal*, 2013. dx.doi.org/doi:10.3888/tmj.15-8.

About the Author

Nikos D. Bagis is a mathematician with a PhD in Mathematical Informatics from Aristotle University of Thessaloniki.

N. D. Bagis
Stenimahou 5 Edessa Pellas
58200 Greece
nikosbagis@hotmail.gr